Chapter 13

Vector Analysis

The physical sciences are rich with examples of interesting problems, which are solved through the method of vector analysis. This topic brings together much of what we have learned about differentiation, integration, and vectors. The concepts involved are fairly straight forward, given our past experiences in calculus, but the ideas quickly lead to problems which are computationally intensive. Maple is an excellent environment to solve the problems of vector analysis. The Maple commands that we are already familiar with are all we need, and so we can focus all of our attention on the mathematical ideas involved.

Since this book is supplementary, we defer to your main text to develop the ideas of vector analysis more carefully. The language used here is intended to foster an intuitive understanding of the ideas.

1 Path and Line Integrals

Consider a curve C defined parametrically by a differentiable, vector valued function $\vec{r}(t) = x(t)\vec{i} + y(t)\vec{j} + z(t)\vec{k}$, for t in some interval $I = [a, b]$. Think of t as time, and $\vec{r}(t)$ as the position vector of an object moving along the curve. Recall that in this context, a position vector, $\vec{r}(t)$, is always based at the origin (so that its components are the coordinates of its head), and the position of the object on the curve C at time t is determined by the head of the position vector. From earlier chapters, we know that the vector $\vec{v}(t) = \vec{r}\,'(t) = x'(t)\vec{i} + y'(t)\vec{j} + z'(t)\vec{k}$ is the velocity of the object and $|\vec{v}(t)|$ is its speed at time t.

The formula $\vec{r}(t)$, which describes the curve C is called a parametrization of C. There are many different parametrizations, which describe the same curve C. A parametrization is, after all, just a particular way of moving along the curve. The notion of orientation of a curve is difficult to pin down without becoming abstract. Your main text will discuss this in more detail. If we exclude back and forth movement along C, then basically, we can select one direction along the curve as forward motion, and the other direction as backward motion. This gives us the two orientations of the curve.

In vector analysis, **we assume that the parametric representation, $\vec{r}(t)$ $(a \le t \le b)$, of a curve is continuously differentiable** ($\vec{v}(t) = \vec{r}\,'(t)$ is continuous). This assumption is natural—$\vec{v}(t)$ appears in all of the integrals. **We also assume that $\vec{v}(t)$ is never the zero vector**. This has many advantages. It implies the **existence of a continuously turning unit tangent vector**,

$$\vec{T}(t) = \frac{1}{|\vec{v}(t)|}\vec{v}(t),$$

at each point on the curve. It **prevents troublesome, sharp corners** from appearing on the curve. It **prevents back and forth motion** along the curve. We will call **$\vec{r}(t)$ a smooth parametric representation** of a curve C if $\vec{r}(t)$ is **continuously differentiable with a derivative which is nowhere zero**.

Mathematicians typically use the symbol $s(t)$ to denote the length of a curve, $\vec{r}(t)$ $(a \le t \le b)$, from the initial point $\vec{r}(a)$ on the curve to the point $\vec{r}(t)$. We will not pursue this idea, but it may explain why the symbol ds is used to denote a small difference in arc length, or, in other words, the length of a small subarc.

Let $\vec{r}(t)$ $(a \le t \le b)$ be a smooth parametric representation of a curve C, with $\vec{v}(t) = \vec{r}\,'(t) \neq \vec{)}$ for all t. Let $f(x, y, z)$ be a real valued funtion, which is defined and continuous at points $P(x, y, z)$ on the curve C (at least). The **path integral of f along C** is evaluated by the formula

$$\int_C f\,ds = \int_a^b f(x(t), y(t), z(t))|\vec{v}(t)|\,dt.$$

In some books, this serves as a definition as well. It is important that the values of a, and b, which determine the parametric interval, satisfy $a < b$. The orientation of the curve can always be controled by the parametric representation, rather than by letting t go backwards on some interval. For example, if an oriented curve C is defined by $\vec{r}(t)$ for t in the interval $0 \le t \le 1$, then $\vec{r}(1-t)$ for $0 \le t \le 1$ describes the same curve with the opposite orientation. In both cases the parameter t goes from 0 to 1.

This formula for a path integral carries an immediate interpretation. Think of t as a moment in time, and dt as a small increment of time. If dt is small, speed, $|\vec{v}(t)|$ is almost constant. Therefore, $ds = |\vec{v}(t)|\,dt$ is almost the length of the subarc of C traversed during this short time increment at t. The term $f\,ds$ represents the value of f at some point in the small subarc times the length ds of the subarc. Finally, we can think of $\int_C f\,ds$ as the sum (in the sense of integration) of all the terms $f\,ds$ as t goes from a to b.

In this sense, the path integral looks very much like the elementary integral, $\int_a^b f(x)\,dx$, studied near the beginning of your calculus program. The resemblance is intentional, but **there is one very important difference**. In the elementary integal, the symbol dx, represents a directed difference in x-values. We think of dx as a positive difference in x-values, if $a < b$ (x is increasing from a to b), and as a negative difference, if $a > b$ (x is decreasing from a to b).

Contrast this interpretation of dx in the elementary integral with the interpretation of $ds = |\vec{v}(t)|\,dt$ in a path integral. **Notice that ds is always positive**, regardless of whether the curve is traversed from one end to the other or vice versa. This suggests that the value of a line integral should not depend on the orientation of the curve. Your main text will state a result which says that **path integrals are not only independent of orientation, but independent of parametrization as well.** A path integral is said to be an **unoriented integral**. No mystery is involved here. We would not expect the sum of the terms $f ds$ to depend on how slowly or quickly we add them up, or whether we add them up from one end to the other, or vice versa.

This interpretation of a path integral, which is the basis for all of its applications, can be placed in a more rigorous setting by alternately defining a path integral in the form

$$\int_C f\,ds = \lim_{\Delta t \to 0} \sum_{j=1}^{n} f(x(t_j), y(t_j), z(t_j))|\vec{v}(t_j)|\Delta t_j.$$

A problem of approximating a path integral, using this alternate Riemann sum definition appears in the exercises.

Example 1 *Evaluate the path integral of $f(x,y,z) = 5y(x^2z-2)^2$ over the spiral curve C defined by $\vec{r}(t) = 2\cos(t)\vec{i} + 2\sin(t)\vec{j} + 5t\vec{k}$ $(0 \le t \le 4\pi)$ The parametrization $\vec{R}(t) = \vec{r}(4\pi - t)$ for $0 \le t \le 4\pi$ describes the same curve with the opposite orientation. Verify that the path integral obtained using this parametrization is the same as the original value.*

The function $f(x,y,z)$ is clearly continuous everywhere. The formula for the velocity vector, shown in the following Maple work session, clearly shows that $\vec{r}(t)$ is a smooth parametric representation.

****

```
> with(linalg):

Warning, new definition for norm

Warning, new definition for trace
> r:=vector([2*cos(t),2*sin(t),5*t]);
```

$$r := [2\cos(t),\ 2\sin(t),\ 5\,t]$$

```
> f:=(x,y,z)->5*y*(x^2*z-2)^2;
```

$$f := (x,\ y,\ z) \to 5\,y\,(x^2\,z-2)^2$$

```
> with(plots):
> spacecurve(r,t=0..4*Pi);
> v:=map(diff,r,t);
```

$$v := [-2\sin(t),\ 2\cos(t),\ 5]$$

Speed is the length of the velocity vector, which could be expressed as norm(v,2). Recall, however, that Maple has occasional difficulty working with the



absolute value bars that enter into expressions involving the norm( ) command. To avoid this problem, we set up our own length expression.#

```
> speed:=sqrt(v[1]^2+v[2]^2+v[3]^2);
```

$$speed := \sqrt{4\sin(t)^2 + 4\cos(t)^2 + 25}$$

```
> speed:=simplify(speed);
```

$$speed := \sqrt{29}$$

The path integral can now be computed directly, but we set it up using the inert form of the integral command, so that we can view its structure.#

```
> PathIntegral:=Int(f(r[1],r[2],r[3])*speed,t=0
> ..4*Pi);
```

$$PathIntegral := \int_0^{4\pi} 10\sin(t)\,(20\cos(t)^2\,t - 2)^2\,\sqrt{29}\,dt$$

```
> answer:=value(PathIntegral);
```

$$answer := -12800\sqrt{29}\,\pi^2 + \frac{3200}{3}\sqrt{29}\,\pi$$

```
> evalf(answer);
```

$$-662267.0521$$

```
> R:=subs(t=4*Pi-t,evalm(r));
```

$$R := [2\cos(4\pi - t),\ 2\sin(4\pi - t),\ 20\pi - 5t]$$

```
> v:=map(diff,R,t);
```

$$v := [-2\sin(t),\ -2\cos(t),\ -5]$$

```
> speed:=simplify(sqrt(v[1]^2+v[2]^2+v[3]^2));
```

$$speed := \sqrt{29}$$

```
> PathIntegral:=int(f(R[1],R[2],R[3])*speed,t=0
> ..4*Pi);
```

$$PathIntegral := -12800\sqrt{29}\,\pi^2 + \frac{3200}{3}\sqrt{29}\,\pi$$

****

Before we discuss the second major topic of this section, we turn our attention, briefly, to an important related issue.

A (3-dimensional) **vector field is a function, $\vec{F}$, which associates with each point, $P(x, y, z)$, a 3-dimensional vector, $\vec{F}(x, y, z)$ based at P.** (A 2-dimensional vector field would base a 2-dimensional vector $\vec{F}(x, y)$ at each point $P(x, y)$.) Gradient fields, which were discussed back in Section 11.4, are examples of vector fields, but many fields are not gradient fields. Think of a vector field as the velocity flow of a gas, for example. While there are many other useful intepretations, in this section, it helps to think of a vector field as a **force field**. Maple has several **commands** in the plots package for **plotting fields**. We used the gradplot( ) comand in Section 11.4. The 3-dimensional commands may not

be very useful, but the 2-dimensional versions can occasionally provide much needed insight into a problem. Here is a simple example.

****

```
> with(linalg):

Warning, new definition for norm

Warning, new definition for trace
> with(plots):
> F:=vector([-y,x]);
```

$$F := [-y,\ x]$$

```
> fieldplot(F,x=-10..10,y=-10..10);
```

****

We are ready to introduce the notion of a line integral. Notice, right from the start, that it seems to depend, in contrast to a path integral, on the orientation of the curve.

Suppose that $\vec{F}(x, y, z)$ is a continuous vector field. Let $\vec{r}(t) = x(t)\vec{i}+y(t)\vec{j}+z(t)\vec{k}$ $(a \le t \le b)$ be a smooth parametric representation of a curve C, and let $\vec{v}(t) = \vec{r}\,'(t)$



denote its velocity vector. **The line integral of $\vec{F}(t)$ along the oriented path C** is evaluated (defined) by

$$\int_C \vec{F} \cdot d\vec{s} = \int_a^b \left(\vec{F}(x(t), y(t), z(t)) \cdot \vec{v}(t)\right)\, dt.$$

Notice the symbol $d\vec{s}$ appearing in the above statement, and compare this to the symbol ds used in the formula for a path integral. The symbol, ds, is a scalar quantity representing a small difference in arc length, or, in other words, the length of a small subarc. The symbol, $d\vec{s} = \vec{v}(t)\, dt$, is a vector quantity, which represents the vector from the initial point to the terminal point on a small subarc. Conceptually, we could say that $ds = |d\vec{s}| = |\vec{v}(t)|dt$.

Let $\vec{T}(t)$ denote the unit tangent vector

$$\vec{T}(t) = \frac{1}{|\vec{v}(t)|}\vec{v}(t),$$

to the curve C at the point $\vec{r}(t)$ on C. Then the line integral can be expressed as the path integral

$$\int_C \vec{F}\, d\vec{s} = \int_a^b \left(\vec{F}(x(t), y(t), z(t)) \cdot \frac{\vec{v}(t)}{|\vec{v}(t)|}\right) |\vec{v}(t)|dt = \int_C \left(\vec{F} \cdot \vec{T}\right)\, ds.$$



A physical interpretation of this integral will help. Think of t as a moment in time, and $\vec{F}$ as a force field. Then $\vec{F} \cdot \vec{T}$ represents the component of force in the direction of the unit tangent vector $\vec{T}$, as shown in the diagram. Over a small time increment, dt, the point on the curve moves a small distance of ds in the direction of $\vec{T}$ under the influence of an almost constant force $\vec{F} \cdot \vec{T}$. The work or energy associated with this movement is the product of force times distance, or $\left(\vec{F} \cdot \vec{T}\right) ds$. To find the total work required to move the object from the initial point on the curve to its terminal point, we sum, in the sense of integration, all of the incremental work terms along the curve.

If the angle between $\vec{F}$ and $\vec{T}$ is acute, as it is in the picture, then $\vec{F} \cdot \vec{T}$ will be positive. The force is pushing the object forward. **A positive term means that energy is released as the object moves forward.**

If the angle between $\vec{F}$ and $\vec{T}$ is obtuse, then $\vec{F} \cdot \vec{T}$ will be negative. The force is pushing against the object. **A negative term means that energy must be put into the system to move the object forward.**

The term $\vec{F} \cdot \vec{T}$ depends strongly on the orientation of the curve. Unit tangent vectors based at the same point of the curve but coming from oppositely oriented parametrizations of C are clearly negatives of each other, and so their dot products with $\vec{F}$ are also negatives of each other. If C^+, and C^- are oppositely oriented representations of the same curve, then

$$\int_{C^+} \vec{F}\, d\vec{s} = -\int_{C^-} \vec{F}\, d\vec{s}.$$

Line integrals are said to be oriented integrals. The value of $\int_C \vec{F}\, d\vec{s}$ is the same for any two smooth parametrizations of C with the same orientation. These results will not be proven, but we show, by example, that the results seem to hold.

Example 2 *Let $\vec{F}$ be the vector field*

$$\vec{F}(x, y, z) = x^2yz\vec{i} + y^2z^2\vec{j} + (x^3 + z^2)\vec{k},$$

and let C be the curve from $P(-1, -1, 1)$ to $Q(1, 1, 1)$ defined by

$$\vec{r}(t) = t^3\vec{i} + t\vec{j} + t^2\vec{k} \ (-1 \le t \le 1).$$

Determine the line integral of $\vec{F}$ along C. The formula,

$$\vec{r}_1(t) = \vec{r}(p(t)) \ (-1 \le t \le 1) \ \textit{for} \ p(t) = \frac{t^2 + 3t - 1}{3},$$

is another smooth parametric representation of the same curve with the same orientation. Verify that the value of the line integral is the same using this representation. The formula,

$$\vec{r}_2(t) = \vec{r}(1 - 2t) \ (0 \le t \le 1),$$

is a smooth parametric representation for the same curve with the opposite orientation. Verify that its line integral is the negative of the original.

By looking at the velocity formulas in the following work session it is clear that all three parametric representations are smooth.

Why is $\vec{r}_1(t)$ another parametric representation of C? Notice that $p(-1) = -1$, $p(1) = 1$, and that $p(t)$ is an increasing function on [-1,1] (its derivative is positive). Thus $p(t)$ maps the interval [-1,1] onto [-1,1]. Think of $t^* = p(t)$ as just another t-number in the interval [-1,1], and it will be clear that $\vec{r}_1(t) = \vec{r}(t^*)$ traces out the same path C. The argument is much the same for $\vec{r}_3(t)$.

****

```
> with(linalg):

Warning, new definition for norm

Warning, new definition for trace
> F:=(x,y,z)->vector([x^2*y*z,y^2*z^2,x^3+z^2])
> ;
```

$$F := (x,\, y,\, z) \to \mathrm{vector}([x^2\, y\, z,\, y^2\, z^2,\, x^3 + z^2])$$

```
> r:=vector([t^3,t,t^2]);
```

$$r := [t^3,\, t,\, t^2]$$

```
> v:=map(diff,r,t);
```

$$v := [3\, t^2,\, 1,\, 2\, t]$$

```
> G:=dotprod(F(r[1],r[2],r[3]),v);
```

$$G := 3\, t^{11} + t^6 + 2\,(t^9 + t^4)\, t$$

```
> LineIntegral:=int(G,t=-1..1);
```

$$\mathit{LineIntegral} := \frac{50}{77}$$

```
> r1:=subs(t=(t^2+3*t-1)/3,evalm(r));
```

$$\mathit{r1} := \left[(\frac{1}{3}\, t^2 + t - \frac{1}{3})^3,\, \frac{1}{3}\, t^2 + t - \frac{1}{3},\, (\frac{1}{3}\, t^2 + t - \frac{1}{3})^2\right]$$

```
> v1:=map(diff,r1,t);
```

$$\mathit{v1} := \left[3\,(\frac{1}{3}\, t^2 + t - \frac{1}{3})^2\,(\frac{2}{3}\, t + 1),\, \frac{2}{3}\, t + 1,\, 2\,(\frac{1}{3}\, t^2 + t - \frac{1}{3})\,(\frac{2}{3}\, t + 1)\right]$$

```
> LineIntegral2:=int(dotprod(F(r1[1],r1[2],r1[3
> ]),v1),t=-1..1);
```

$$LineIntegral2 := \frac{50}{77}$$

```
> r2:=subs(t=1-2*t,evalm(r));
```

$$r2 := \left[(1-2\,t)^3,\; 1-2\,t,\; (1-2\,t)^2\right]$$

```
> v2:=map(diff,r2,t);
```

$$v2 := \left[-6\,(1-2\,t)^2,\; -2,\; -4+8\,t\right]$$

```
> LineIntegral3:=int(dotprod(F(r2[1],r2[2],r2[3
> ]),v2),t=0..1);
```

$$LineIntegral3 := \frac{-50}{77}$$

****

Example 3 *A playground slide has the shape of the curve*

$$\vec{r}(t) = 5\cos(t)\vec{\imath} + 5\sin(t)\vec{\jmath} + (2 + t^2/4)\vec{k},$$

where the units are in feet. Assume all of its mass is concentrated at points on the curve itself. In order to carry the weight of the upper portion of the slide, the mass density increases as elevation decreases. The mass density, in pounds per foot, at elevation z is $\delta(z) = 45 - 4z$. Determine the mass, and center of mass of the slide.

Imagine the slide (the curve) partitioned into small subarcs, so that each subarc can be treated almost as a point mass. We use small case and capital letters to denote, in a fairly predictable fashion, the various subterms involved. Thus, for example, m denotes the mass of a small subarc, and M denotes the sum (in the sense of integration) of all the masses of the subarcs. In the same fashion, mxy, Mxy denote the individual and total moments with respect to the xy-coordinate plane.

Actual time is, of course, not involved in this problem, but it is a common and convenient name for a parameter and so we shall use it. The length of a subarc, which we denote by $SubarcSize$ in the following Maple work session, should involve the "time increment", dt. The differential symbol, however, is assumed by the integration command, and so it does not appear in any formula. Conceptually, the assignment should be, $SubarcSize := speed\,dt$, but the dt is dropped because of this.

****

```
> with(linalg):

Warning, new definition for norm
```

```
Warning, new definition for trace
> r:=vector([5*cos(t),5*sin(t),2+t^2/4]);
```

$$r := \left[5\cos(t),\ 5\sin(t),\ 2+\frac{1}{4}t^2\right]$$

```
> delta:=z->45-4*z;
```

$$\delta := z \to 45 - 4\,z$$

```
> with(plots):
```

#This plot includes some changes made by clicking the options box in the plot window. In particular, **Constrained** was selected from the **Projection** menu.#

```
> spacecurve(r,t=0..6);
```




```
> v:=map(diff,r,t);
```

$$v := \left[-5\sin(t),\ 5\cos(t),\ \frac{1}{2}t\right]$$

```
> speed:=sqrt(v[1]^2+v[2]^2+v[3]^2);
```

$$speed := \frac{1}{2}\sqrt{100\sin(t)^2 + 100\cos(t)^2 + t^2}$$

```
> SubarcSize:=simplify(speed);
```

$$SubarcSize := \frac{1}{2}\sqrt{100+t^2}$$

```
> m:=delta(r[3])*SubarcSize;
```

$$m := \frac{1}{2}(37-t^2)\sqrt{100+t^2}$$

```
> myz:=r[1]*m;
```

$$myz := \frac{5}{2}\cos(t)\,(37 - t^2)\,\sqrt{100 + t^2}$$

```
> mxz:=r[2]*m;
```

$$mxz := \frac{5}{2}\sin(t)\,(37 - t^2)\,\sqrt{100 + t^2}$$

```
> mxy:=r[3]*m;
```

$$mxy := \frac{1}{2}\,(2 + \frac{1}{4}\,t^2)\,(37 - t^2)\,\sqrt{100 + t^2}$$

```
> M:=evalf(Int(m,t=0..5));
```

$$M := 738.8906164$$

```
> Myz:=evalf(Int(myz,t=0..5));
```

$$Myz := -465.4351541$$

```
> Mxz:=evalf(Int(mxz,t=0..5));
```

$$Mxz := 1088.588369$$

```
> Mxy:=evalf(Int(mxy,t=0..5));
```

$$Mxy := 2695.580330$$

```
> [Myz/M,Mxz/M,Mxy/M];
```

$$[-.6299107659,\ 1.473274047,\ 3.648145301]$$

****

Example 4 *A two dimensional force field is defined by* $\vec{F}(x,y) = x * y\vec{i} + (x^2 - y^2)\vec{j}$. *Determine the work required to move an object along the spiral curve* $\vec{r}(t) = (2 + t)\cos(2t)\vec{i} + (2 + t)\sin(2t)\vec{j}$ $(0 \le t \le 5)$.

Although an answer is readily available without a graph, it will mean more, intuitively, if we plot the force field and the curve. **Notice how the display( ) command is used to plot both of these together**. A full colon is used to suppress unwanted output in the assignments $p1$, and $p2$ below.

****

```
> with(plots):
> with(linalg):

Warning, new definition for norm

Warning, new definition for trace
> F:=(x,y)->vector([y*x,x^2-y^2]);
```

$$F := (x,\, y) \rightarrow \mathrm{vector}([y\,x,\, x^2 - y^2])$$

```
> r:=vector([(2+t)*cos(2*t),(2+t)*sin(2*t)]);
```

$$r := [(2+t)\cos(2\,t),\, (2+t)\sin(2\,t)]$$

```
> p1:=fieldplot(F(x,y),x=-10..10,y=-10..10):
> p2:=plot([r[1],r[2],t=0..5]):
> display({p1,p2});
```



```
> w:=dotprod(F(r[1],r[2]),r);
```

$$w := (2+t)^3 \sin(2\,t)\cos(2\,t)^2 + (4\cos(2\,t)^2 + 4\cos(2\,t)^2\,t + \cos(2\,t)^2\,t^2 - 4\sin(2\,t)^2 - 4\sin(2\,t)^2\,t - \sin(2\,t)^2\,t^2)\,(2+t)\sin(2\,t)$$

```
> Work:=evalf(int(w,t=0..5));
```

$$Work := -39.20683726$$

```
Warning: new definition for   norm
Warning: new definition for   trace
```

****

The negative answer means that energy must be put into the system to move the object in this way.

2 Green's Theorem

The notation for a line integral has another form, which is worth discussing. Let

$$\vec{r}(t) = x(t)\vec{i} + y(t)\vec{j} + z(t)\vec{k} \ (a \le t \le b)$$

be a smooth parametric representation of a curve C, and let

$$\vec{F}(x,y,z) = P(x,y,z)\vec{i} + Q(x,y,z)\vec{j} + R(x,y,z)\vec{k}$$

be a continuous vector field. Then

$$\int_C \vec{F} \cdot d\vec{s} = \int_a^b \left(\vec{F}(x(t), y(t), z(t)) \cdot \vec{v}(t) \right) dt$$
$$= \int_a^b \left(P(x,y,z)\tfrac{dx}{dt} + Q(x,y,z)\tfrac{dy}{dt} + R(x,y,z)\tfrac{dz}{dt} \right) dt$$

By allowing the symbols dt to cancel, we obtain the alternate notation

$$\int_C \vec{F} \cdot d\vec{s} = \int_C (P\,dx + Q\,dy + R\,dz)$$

for a line integral. This is probably the most common notation for a line integral, although it has no computational advantage. It emphasizes the important point that the value of a line integral is independent of the parameterization (except for orientation). It emphasizes the variables x, y, and z, and omits the parameter t. It suggests that we think of a curve, C, purely as a static set of points (x,y,z) along with an arrow on the curve indicating its orientation.

A curve, C is said to be rectifiable, if it has finite arc length. This definition is a case in point! No mention is made of the parameter. Now, it is easy to see, that if $\vec{r}(t) = x(t)\vec{i} + y(t)\vec{j} + z(t)\vec{k}$ $(a \le t \le b)$ is a smooth representation of C, then the curve is rectifiable. An elementary formula for arc length is

$$s = \int_a^b |\vec{v}(t)|\, dt = \int_a^b \sqrt{(x'(t))^2 + (y'(t))^2 + (z'(t))^2} dt.$$

For a smooth representation, $\vec{v}(t)$ is continuous, and so this integral exists.

In a two dimensional problem, $R = 0$, $z = 0$, and the notation for a line integral becomes

$$\int_C \vec{F} \cdot d\vec{s} = \int_C (P\,dx + Q\,dy)\,.$$

It is this two dimensional form of the line integral that will occupy our attention in the remainder of this section.

A curve, C, in the plane is said to be **simple, if it does not intersect itself, except possible at its end points**. It is a **closed curve, if its end points are the same**. If a curve is both simple and closed, its orientation can be described as being either counterclockwise, or clockwise. The **counterclockwise (positive) orientation on C is denoted by C^+, and the clockwise (negative) orientation on C is denoted by C^-**. The sign convention is arbitrary, but it is consistent with the sign usually attached to an angle in a rotation.

The symbol **∂D is used to denote the boundary of a region D** in the plane.

Green's Theorem states, that

$$\int_{\partial D+} (P\,dx + Q\,dy) = \int\int_D \left(\frac{\partial Q}{\partial x} - \frac{\partial P}{\partial y}\right)\,dxdy,$$

as long as $\vec{F} = P\vec{i} + Q\vec{j}$ is continuously differentiable on a region D whose boundary, ∂D, is a simple closed rectifiable curve.

This result, and its 3-dimensional counterpart, *Stokes Theorem*, are among the most important results in all of calculus.

Hand calculations can often be simplified, substantially, by electing (regardless of an integral's initial form) to do the line integral, or the double integral, which ever is simpler.

From a Maple point of view this may not be very impressive. After all, we may seldom care how hard we make Maple work for an answer. In spite of this, *Green's Theorem*, and *Stoke's Theorem* rank side-by-side with the *Fundamental Theorem of Calculus* in order of importance. The power of these results comes, not from its calculation, but from its various interpretations to pure mathematics, and to many applied fields.

Example 5 *Verify* Green's Theorem *for* $P(x,y) = x^3 + 5x^2y^2 + y^3x$*,* $Q(x,y) = 3x + 5y^2 + 1$*, where* D *is the elliptical region*

$$\frac{(x-4)^2}{2^2} + \frac{(y-7)^2}{4^2} \leq 1.$$

The standard way of parametrizing the bounding ellipse is to use the identity $\cos^2(t) + \sin^2(t) = 1$. We set $(x-4)/2 = \cos(t)$ and $(y-7)/7 = \sin(t)$, which defines the ellipse parametrically in terms of t.

****

```
> P:=(x,y)->x^3+5*x^2*y^2+y^3*x;Q:=(x,y)->3*x+5*y^2+1;
```

$$P := (x,\, y) \to x^3 + 5\,x^2\,y^2 + y^3\,x$$

$$Q := (x,\, y) \to 3\,x + 5\,y^2 + 1$$

```
> x:=4+2*cos(t);y:=7+4*sin(t);
```

$$x := 4 + 2\cos(t)$$

$$y := 7 + 4\sin(t)$$

```
> F1:=P(x,y)*diff(x,t)+Q(x,y)*diff(y,t);
```

$$F1 := -2((4+2\cos(t))^3 + 5\,(4+2\cos(t))^2\,(7+4\sin(t))^2 + (7+4\sin(t))^3\,(4+2\cos(t)))$$
$$\sin(t) + 4\,(13 + 6\cos(t) + 5\,(7+4\sin(t))^2)\cos(t)$$

```
> int(F1,t=0..2*Pi);
```

$$-14584\,\pi$$

```
> x:='x':     y:='y':
> F2:=diff(Q(x,y),x)-diff(P(x,y),y);
```

$$F2 := 3 - 10\,x^2\,y - 3\,y^2\,x$$

```
> eq:=(x-4)^2/4+(y-7)^2/16=1;
```

$$eq := \frac{1}{4}(x-4)^2 + \frac{1}{16}(y-7)^2 = 1$$

```
> Y:=solve(eq,y);
```

$$Y := 7 + 2\sqrt{-12 - x^2 + 8\,x},\ 7 - 2\sqrt{-12 - x^2 + 8\,x}$$

```
> int(int(F,y=Y[2]..Y[1]),x=2..6);
```

$$-14584\,\pi$$

****

It follows from *Green's Theorem* that the area of a region D in the plane satisfies

$$\text{Area} = \frac{1}{2}\left(\int_{\partial D} x\,dy - \int_{\partial D} y\,dx\right).$$

Example 6 *Let D be the region bounded by the curves*

$$x = y^4 - 5y^3 + 2y^2 - 13; x = 57 + 10y + \frac{y}{2}.$$

Compute the area of D directly, and by using Green's Theorem*.*

****

```
> eq1:=x=y^4-5*y^3+2*y^2-13;eq2:=x=57+10*y+y/2;
```

$$eq1 := x = y^4 - 5\,y^3 + 2\,y^2 - 13$$

$$eq2 := x = 57 + \frac{21}{2}\,y$$

```
> with(plots):
> implicitplot({eq1,eq2},x=-60..150,y=-4..10);
```




```
> f:=unapply(rhs(eq1),y);g:=unapply(rhs(eq2),y);
```

$$f := y \to y^4 - 5\,y^3 + 2\,y^2 - 13$$

$$g := y \to 57 + \frac{21}{2}\,y$$

```
> a:=fsolve(f(y)=g(y),y=-3..-1);b:=fsolve(f(y)=g(y),y,y=4..6);
```

$$a := -1.853782082$$

$$b := 5.426176335$$

```
> A1:int(int(1,x=f(y)..g(y)),y=a..b);
```

$$659.0724332$$

```
> intxdy:=int(g(y),y=a..b)+int(f(y),y=b..a);
```

$$intxdy := 659.0724332$$

```
> intydx:=int(y*diff(g(y),y),y=a..b)+int(y*diff(f(y),y),y=b..a);
```

$$intydx := -659.0724348$$

```
> A2:=(intxdy-intydx)/2;
```

$$A2 := 659.0724340$$

---

* *